
Introduction

This document explains the properties and theory behind random number generation in more detail.  

We presume that you have read Section 9.2 Random Number Generation Theory.  Please see the references at the end of the chapter for more documentation and suggestions for further research.
The explanation below contains references to Microsoft’s Knowledge Base (KB), which is available at support.microsoft.com/.  This excellent resource goes way beyond Excel’s Help and features detailed documentation and bug reports.  Articles are referenced by a KB Article ID, and the entire archive is searchable.
In addition to documentation on RAND, explained more fully below, we recommend these KB Articles:
Visual Basic’s RND Function: KB 231847

VB’s RND function is NOT the same as Excel’s RAND function.

Excel 2003 Statistical Functions Overview: KB 828888

Excel 2003 saw a major upgrade and revision in a variety of statistical functions.

Excel 2003 LINEST Function: KB 828533

Documents corrections made in LINEST for Excel 2003

1. Excel’s RAND (before Excel2003) in Detail

Excel’s RAND function is a poor RNG.  It has an unknown period and successive pairs show a strong pattern.  It is surprising that it does as well as it does in Marsaglia’s DIEHARD battery of tests.

As described in section 9.2, Random Number Generation Theory, RAND is an LCG: 9821*x-1 + 0.211327.  However, matters  immediately become complicated because you cannot replicate RAND in a spreadsheet.
All of the cells in Excel are declared as Variants.  This enables Excel to handle numbers and strings (text) in cells.  Unfortunately, this also means that you cannot replicate RAND on a spreadsheet in Windows.  When calculating with Variants, Excel employs coercion rules to enable the computation. This causes a slight difference in the computed value with a spreadsheet versus RAND itself (which is computed within the core Excel.exe program).

Amazingly, you can replicate RAND in Mac Excel because it appears that Mac Excel VBA and core Excel.exe use different coercion rules!  We ignore Mac Excel in the remainder of this document because its RAND function, though slightly different, behaves similarly to its Windows cousin.

From Windows Excel, open the workbook RAND.xls.

The workbook demonstrates two important facts:

1. Excel’s RAND cannot be replicated from within a spreadsheet.

2. Excel’s RAND can be replicated by Visual Basic.
Open the RANDBehavior.xls workbook.
Had Microsoft coded the RAND LCG so that it exactly computed the next value without any precision error, then RAND would behave as expected.  The Testing sheet shows that, by using the Decimal data type, you can compute the next value exactly.
The Exact sheet shows that its period would be exactly 1,000,000 (as reported in Microsoft’s documentation, “This formula will provide up to 1 million different numbers.” (KB Article 86523).  The successive pairs test would show that there are just two lines in the 0.7 to 0.7001 interval visited by this LCG.  
We did not bother to subject this LCG to the DIEHARD battery of tests because it seems rather obvious that it would fail spectacularly.  Moreover, this is not what Excel’s RAND is actually doing.
In fact, Excel’s RAND is not simply 9821*x-1 + 0.211327 because of floating-point computer arithmetic issues and the data type used.  So what are the properties of an algorithm that could be called “9821*x-1 + 0.211327 with some floating-point precision error propagating through each iteration?”
First, the exact period cannot be determined via numerical testing.  RAND simply doesn’t revisit a number even after running for a week.  The fact that it that has a long (of unknown length) period is not enough to make it a good RNG.  The Double sheet shows that it suffers from the same high-order structure in the successive-pairs test as the pure LCG 9821*x-1 + 0.211327.
Pierre L’Ecuyer concludes a paper on testing various RNGs with a warning and advice:
Do not trust the random number generators provided in popular commercial software such as Excel, Visual Basic, etc, for serious applications.  Some of these RNGs give totally wrong answers for the two simple simulation problems considered in this paper. Much better RNG tools are now available, as we have just explained in this paper.  Use them.  If reliable RNGs are not available in your favorite software products, tell the vendors and insist that this is a very important issue.  An expensive house built on shaky foundations is a shaky house.  This applies to expensive simulations as well.
 
2. Excel2003 RAND

With the release of Excel2003 in October of that year, Microsoft Corporation coded a new RAND function.  They went with Wichman and Hill’s AS 183 three-stream RNG (1982) and documented the routine in the Knowledge Base with article title, “Excel Statistical Functions RAND” (KB 828795).
As the documentation explains, 
the RAND function in earlier versions of Excel used a pseudo-random number generation algorithm whose performance on standard tests of randomness was not sufficient. Although this is likely to affect only those users who have to make a large number of calls to RAND, such as a million or more, and not to be a concern for almost every user, the pseudo-random number generation algorithm that is described here was implemented for Excel 2003. It passes the same battery of standard tests.

The battery of tests is named Diehard (see note 1). The algorithm that is implemented in Excel 2003 was developed by B.A. Wichman and I.D. Hill (see note 2 and note 3). This random number generator is also used in the RAT-STATS software package that is provided by the Office of the Inspector General, U.S. Department of Health and Human Services. It has been shown by Rotz et al (see note 4) to pass the DIEHARD tests and additional tests developed by the National Institute of Standards and Technology (NIST, formerly National Bureau of Standards).
See More Information in KB 828795
Unfortunately, within days of the release, word got out about a bug in the new RAND.  The new RAND, as implemented in Excel2003, yields negative numbers.

If you have the original release of Excel2003, you can replicate this behavior by opening Excel2003RANDBug.xls.  Instructions are provided that show you how to generate negative numbers with RAND.  

On January 29, 2004, Microsoft released a Hotfix Package that corrects the problem.  The bug and a link to the patch are available in KB 834520.

We have not tested Excel2003 RAND (either the original or patched versions) as of this writing, but, needless to say, this behavior does not inspire confidence.  We continue to recommend using our implementation of FMRG via the RANDOM() function.
3. Barreto/Howland’s Implementation of FMRG

The RANDOM() function in the MCSim add-in and which is used in many of our workbooks is entirely based on Deng, Lih-Yuan and Dennis K. J. Lin, (2000) “Random Number Generation for the New Century,” The American Statistician, 54(2):145-150. 
We used the Currency data type as our long integer data type because it is longer than a Long and meets the requirements of the algorithm.  The CurrencyDoc sheet in the workbook RANDOM.xls explains how this data type can be used to “fake” a 64-bit integer in Visual Basic.
By placing the code in the auto_open subroutine, Excel executes the code whenever the workbook (or add-in) is opened.  The macros that call FMRG are included below.

You can easily use the RANDOM algorithm by installing the MCSim.xla add-in or by copying the RNGandSortModule to a workbook.

Here is the code from the RNGandSortModule in the RANDOM.xls workbook: 
Public myFMRG As Double

Public myFMRGInt As Currency ' see sheet CurrencyDoc in Random.xls for documentation

Public myFMRGIntlag1 As Currency

Public myFMRGIntlag2 As Currency

Public B As Long

Const p As Long = 2147483647

Sub auto_open()

' Initialize FMRG

Randomize 'Randomize runs on open; no need to use Randomize again in any macro

myFMRGIntlag1 = Rnd * 2 ^ 24

myFMRGIntlag2 = Rnd * 2 ^ 24

'Load B values from p. 147 of Deng and Lin, "Random Number Generation for the New Century," The American Statistician, May 2000, vol. 54, no. 2

Dim BArray(1 To 25) As Long

BArray(1) = 26403

BArray(2) = 27149

BArray(3) = 29812

BArray(4) = 30229

BArray(5) = 31332

BArray(6) = 33236

BArray(7) = 33986

BArray(8) = 34601

BArray(9) = 36098

BArray(10) = 36181

BArray(11) = 36673

BArray(12) = 36848

BArray(13) = 37097

BArray(14) = 37877

BArray(15) = 39613

BArray(16) = 40851

BArray(17) = 40961

BArray(18) = 42174

BArray(19) = 42457

BArray(20) = 43199

BArray(21) = 43693

BArray(22) = 44314

BArray(23) = 44530

BArray(24) = 45670

BArray(25) = 46338

'Pick a B, each with equal probability

Dim myX As Integer

myX = Rnd * 24 + 1

B = BArray(myX)

End Sub

Sub FMRG()

'To use FMRG in VBA code: FMRG runs the rng, myFMRG is the actual random number drawn

'To get a VBA Rnd, just use

'   Rnd

'To get an FMRG value, TWO steps are needed

'   FMRG

'   x = myFMRG to get the random number and put it into x

'If you RESET VB at any time, you clear the value of B

'The If statement below runs auto_open again when B = 0

If B = 0 Then auto_open

'This calculates the next "random" number in the sequence

myFMRGInt = (B * myFMRGIntlag2 - myFMRGIntlag1) - p * Int((B * myFMRGIntlag2 - myFMRGIntlag1) / p)

myFMRG = myFMRGInt / p

myFMRGIntlag2 = myFMRGIntlag1

myFMRGIntlag1 = myFMRGInt

End Sub

Public Function Random() As Double

' To use Random function on a sheet, simply type =random() in a cell

' See sheet1 of Random.xls for an example

Application.Volatile (True)

FMRG

Random = myFMRG

End Function
4. Criticisms of FMRG
Testing a random number algorithm involves much more than simply examining the successive pairs in a small interval as we did in Section 9.2 “Random Number Generation Theory.”
We have used our implementation of Deng and Lin’s FMRG code in a wide variety of situations in which the analytical properties were known either exactly or approximately (i.e., asymptotically).  The algorithm has yet to produce bizarre results.
In May 2003, however, L’Ecuyer and Touzin criticized FMRG:

We study the structure and point out the weaknesses of recently-proposed random number generators based on special types of linear recurrences with small coefficients, which allow fast implementations.  Our theoretical analysis is complemented by the results of simple empirical statistical tests that the generators fail decisively.  
Pierre L’Ecuyer and Renée Touzin, “On the Deng-Lin Random Number Generators and Related Methods,” available online at www.iro.umontreal.ca/~lecuyer/.

L’Ecuyer and Touzin test 5 of the 25 aj coefficients (see BArray in the code in the previous section) and show failures on two tests after 2^21 to 2^25 draws (as determined by the particular aj chosen and the).  Since our implementation of FMRG chooses one of the 25 aj’s via Rnd, it is unclear when FMRG would fail these two particular tests.  

The search for ever faster, better random number generators will continue, and we do not claim that Deng and Lin’s FMRG is perfect.  For teaching purposes in our Monte Carlo simulations, however, it is undoubtedly adequate.  

As with the other software contained in this book, we advise corroboration with other algorithms as a commonsense approach to intelligent decision making.  The more important your project, the more important independent replication becomes.

5. NORMALRANDOM Documentation

We generate normal deviates via the Box-Muller method.  Here is the code, available in the RNGandSortModule in the Random.xls workbook.

Sub NormalRNG(NormRand() As Double, Optional TypeofRand As Boolean, Optional MEAN As Double, Optional SD As Double)

' Numerical Recipes in Fortran 77: The Art of Scientific Computing, 2nd ed.; vol. 1; p. 279

' Box-Muller Method

' To use, call from a macro with FOUR parameter choices

' The first choice requires an array so it has to be DIMmed somewhere in the macro

' Example 1: Explicit Choosing

' Dim result(1 to 100) as double   MUST be DIMmed as double to be compatible with NormalRNG

' NormalRNG result, 0, 0, 1

' These two lines create a result vector that is filled with Normally distributed values based

' on VBA's Rnd

' Example 2: Default Choosing

' Dim result(1 to 100) as double

' NormalRNG result

' Since the default choices are VBA's Rnd, mean zero, and SD = 1, this gives the same result as Example 1

' Example 3:

' Dim myOutput(1 to 100000) as double

' NormalRNG myOutput, 1, 20, 5

' These two lines create a myOutput vector that is filled with Normally distributed values based

' on the FMRG routine

Dim V1 As Double

Dim V2 As Double

Dim Radius As Double

Dim FAC As Double

Dim i As Double

' The Optional parameters are

' TypeofRand, which is 0 or False if not set--this means the macro uses VBA's Rnd

' Getting the number of values requested

Dim number As Double

number = UBound(NormRand())

' Because we generate TWO deviates at once, we divide number in two (with MOD handling odd number)

' We waste one deviate if odd

If TypeofRand = False Then

' Use VBA's Rnd to generate random draws

For i = 1 To number

If ISET = 0 Then

repeat:     V1 = 2 * Rnd - 1

                V2 = 2 * Rnd - 1

                Radius = V1 ^ 2 + V2 ^ 2

    If Radius >= 1 Or Radius = 0 Then GoTo repeat

    FAC = Sqr(-2 * Log(Radius) / Radius)

    GSET = V1 * FAC

    NormRand(i) = V2 * FAC * SD + MEAN

    ISET = 1

Else

    NormRand(i) = GSET * SD + MEAN

    ISET = 0

End If

Next i

Else

' Use FMRG to generate random draws

For i = 1 To number

If ISET = 0 Then

repeat2:   FMRG

                V1 = 2 * myFMRG - 1

                FMRG

                V2 = 2 * myFMRG - 1

                Radius = V1 ^ 2 + V2 ^ 2

    If Radius >= 1 Or Radius = 0 Then GoTo repeat2

    FAC = Sqr(-2 * Log(Radius) / Radius)

    GSET = V1 * FAC

    NormRand(i) = V2 * FAC * SD + MEAN

    ISET = 1

Else

    NormRand(i) = GSET * SD + MEAN

    ISET = 0

End If

Next i

End If

End Sub

'It forces use of FMRG

'Also forces explicit declaration of mean and SD

Function NORMALRANDOM(MEAN As Double, SD As Double) As Double

' To use NormalRandom function on a sheet, simply type =normalrandom() in a cell

' The 3 parameters can be set, eg., =normalrandom(1, 100, 50) uses FMRG, mean 100, SD 50

' See sheet1 of Random.xls for an example

Application.Volatile (True)

Dim NormalRandomValue(1 To 1) As Double

NormalRNG NormalRandomValue, 1, MEAN, SD

NORMALRANDOM = NormalRandomValue(1)

End Function
� “Software for Uniform Random Number Generation: Distinguishing the Good from the Bad”; available online at � HYPERLINK "http://www.iro.umontreal.ca/~lecuyer/" ��www.iro.umontreal.ca/~lecuyer/� and � HYPERLINK "http://www.informs-cs.org/wscpapers.html" ��www.informs-cs.org/wscpapers.html�

